Problem:
Score : 400 points

Problem Statement

We have a large square grid with H rows and W columns.
Iroha is now standing in the top-left cell.
She will repeat going right or down to the adjacent cell, until she reaches the bottom-right cell.

However, she cannot enter the cells in the intersection of the bottom A rows and the leftmost B columns. (That is, there are A×B forbidden cells.) There is no restriction on entering the other cells.

Find the number of ways she can travel to the bottom-right cell.

Since this number can be extremely large, print the number modulo 10^9+7.

Constraints

1 ≦ H, W ≦ 100,000

1 ≦ A < H

1 ≦ B < W

Input

The input is given from Standard Input in the following format:

H W A B

Output

Print the number of ways she can travel to the bottom-right cell, modulo 10^9+7.

Sample Input 1

2 3 1 1

Sample Output 1

2

We have a 2×3 grid, but entering the bottom-left cell is forbidden. The number of ways to travel is two: "Right, Right, Down" and "Right, Down, Right".

Sample Input 2

10 7 3 4

Sample Output 2

3570

There are 12 forbidden cells.

Sample Input 3

100000 100000 99999 99999

Sample Output 3

1

Sample Input 4

100000 100000 44444 55555

Sample Output 4

738162020

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import Data.Bits
import Data.List
import qualified Data.Vector.Unboxed as U
import qualified Data.Vector.Generic as G

modulo :: Int
modulo = 1000000007

(+~) :: Int -> Int -> Int
a +~ b = (a + b) `mod` modulo

(*~) :: Int -> Int -> Int
a *~ b = (a * b) `mod` modulo

powMod :: Int -> Int -> Int
powMod a b = foldl' step 1 [63, 62 .. 0]
  where step x n = let pow2 = x *~ x in
                    if b `testBit` n then pow2 *~ a else pow2

invMod :: Int -> Int
invMod a = powMod (a `mod` modulo) (modulo - 2)

facts :: U.Vector Int
facts = 1 `G.cons` G.scanl1 (*~) (G.enumFromTo 1 200001)

factMod :: Int -> Int
factMod n = facts G.! n

combMod :: Int -> Int -> Int
combMod n r = factMod n *~ invMod (factMod r *~ factMod (n - r))

patterns :: Int -> Int -> Int
patterns h w = combMod (h + w) h

main :: IO ()
main = do
  [h, w, a, b] <- mapM readIO . words =<< getLine
  print $ foldl' (+~) 0 [ patterns row (b - 1) *~ patterns (h - row - 1) (w - b - 1)
                        | row <- [0 .. h - a - 1]
                        ]